****************************************************************************
* Nome do Arquivo: 		codigo89.do
* Data:			 		23/03/2020
* Autor: 				Fred Batista
* Prop�sito: 			An�lise da pesquisa DAT/BR89.SET-00186
* Dados: 				00186.dta
* Output:				-	*/
****************************************************************************

* VARI�VEIS DO QUESTION�RIO NO BANCO

* escala esquerda-direita: q38
* significado de esquerda e direita: q40a-q40e
* regiao: regiao
* estado: estado
* sexo: sexo
* idade: idade
* urbano/rural: area
* voto 10 turno: q1, q2
* partido: partido
* significado de democracia: q23a-q23b
* interesse por pol�tica: q29
* exposi��o � informa��o: q35a-q35c
* le noticiario: q36a
* conversa sore politica: q36b
* trabalha: q42a
* catolico: q45a
* escolaridade: escola
* renda: renda
* confian�a na presid�ncia: q14d
* opiniao sobre o INPS: q20e

* RECODIFICA��O

* opinina��o na escala direita-esquerda

gen loc_resp = q38
recode loc_resp (8=0) (else=1)

* localiza��o na escala direita-esquerda

gen diresq_resp = q38
recode diresq_resp (8=.)
replace diresq_resp = (diresq_resp - 1)/6

* significado de esquerda (0=NS, 1=incorrect/judgment, 2= partial; 3=correct)

gen esq_sig = q40a
recode esq_sig (1=0) (2 3 4 5 6 8 9 10 12 14 15 16 17 19 20=1) (7 13=2) (16 18=3)

* significado de direita (0=NS, 1=incorrect/judgment, 2= partial; 3=correct)

gen dir_sig = q40d
recode dir_sig (1=0) (3 4 5 8 10 12 13 14 15 16 17 18 20=1) (2 7=2) (6 9 11=3)

* sexo

gen homem = sexo
recode homem (1=1) (2=0)

* voto (Lula Brizola Freire a esquerda; Covas Guimaraes Afif Camargo ao centro; Collor Maluf Chaves a direita), base em Carreirao 2002

gen voto1 = q2
recode voto1 (4 5 8= 0) (3 6 10 11=1) (1 2 7=2) (else=.)

* urbano

gen urbano = area
recode urbano (1=1) (2=0)

* partidarismo

gen idpmdb = partido
recode idpmdb (1=1) (else=0)

gen idpt = partido
recode idpt (2=1) (else=0)

gen idpds = partido
recode idpds (3=1) (else=0)

* significado de democracia (0=NS, 1=incorrect/judgment, 2= partial; 3=correct)

gen dem_sig = q23a
recode dem_sig (0 1=0) (7 8 9 13=2) (6 10=3) (2 3 4 57=4)

* conhecimento 

gen conhecimento = (esq_sig + dir_sig + dem_sig)/12

gen conhecimento2 = (esq_sig + dir_sig)/6

* interesse por pol�tica

gen interesse = q29
recode interesse (1 2 3=1) (4 5 6=0)

* le noticias de politica

gen leitura = q36a
recode leitura (1=1) (2 3=0)

* conversa de politica

gen conversa = q36b
recode conversa (1=1) (2 3=0)

* trabalha

gen trabalha = q42a
recode trabalha (1=1) (2=0)

* catolico

gen catolico = q45a
recode catolico (1=1) (2 3=0)

* escolaridade 

gen escolaridade = (escola - 1)/7

* confian�a na presid�ncia

gen confpres = q14d
recode confpres (1=1) (3=.5) (2=0) (4=.)

* opiniao sobre inps

gen inps = q20e
recode inps (6=.)



*** AN�LISES

* testes de validade da medida de conhecimento

reg conhecimento escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

reg conhecimento2 escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

oprobit esq_sig escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

oprobit dir_sig escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

oprobit dem_sig escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

tab conhecimento2 if loc_resp==1

tab esq_sig if loc_resp==1

tab dir_sig if loc_resp==1


* opina��o

probit loc_resp idpmdb idpt idpds escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

margins, dydx(escolaridade)

* coeficiente da escolaridade � o mais alto do modelo

probit loc_resp idpmdb idpt idpds conhecimento escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

probit loc_resp idpmdb idpt idpds conhecimento2 escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico

margins, dydx(conhecimento2 escolaridade)

* coeficiente do conhecimento � o mais alto do modelo e torna o da escolridade nulo


* voto

mprobit voto1 diresq_resp , base(0)

margins, dydx(diresq_resp) predict(outcome(1))

margins, dydx(diresq_resp) predict(outcome(2))


mprobit voto1 diresq_resp conhecimento2 confpres idpmdb idpt idpds escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico, base(0)

margins, dydx(diresq_resp) predict(outcome(1))

margins, dydx(diresq_resp) predict(outcome(2))


mprobit voto1 c.diresq_resp##c.escolaridade confpres idpmdb idpt idpds conhecimento2 interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico, base(0)

margins, dydx(diresq_resp) at(escolaridade=(0(1)1)) predict(outcome(1))

margins, dydx(diresq_resp) at(escolaridade=(0(1)1)) predict(outcome(2))


mprobit voto1 c.diresq_resp##c.conhecimento2 confpres idpmdb idpt idpds escolaridade interesse conversa leitura urbano homem i.renda i.idade i.regiao trabalha catolico, base(0)

margins, dydx(diresq_resp) at(conhecimento2=(0(1)1)) predict(outcome(1))

margins, dydx(diresq_resp) at(conhecimento2=(0(1)1)) predict(outcome(2))
